
Abstract
We develop framework for using genetic algorithm. For library demonstration was implemented
application for extrem computing.
Genetic algorithms
Genetic algorithms are based on behavior of nature. Animals (as example) has genetic code
divided to gens. Gens are inherited from parents. The living entity creates new children. "Better"
entity has more children. So there are prefered gens that has evaluated as "better". One
generations died and other generation grow up. In generation are ontly entities with ability to
survive. That means living entity is better that others. This principle is usefull in genetic
algorithm design.
The framework
In the designed framework are declared three classes
type
TGen = class(TObject)
TGeneration = class(TObject)
TGenetic = class(TObject)
TGen represents one entity in generation. Generation is encapsulated by class TGeneration. This

class (instance of this class) has more instance of TGen. TGenetic is prepared for genetic
algorithm encapsulation. TGeneration is used only for maitenance. Class TGenetic is used for
maitenance too but is posible to extend this class.
All main methods for gen maipulation is declared in class TGen. This function are declared as
virtual so they can be overrided.
type

TGenetic = class;
TGen = class(TObject)
private
FFitness: Double;
procedure SetFitness(const Value: Double);
public
constructor Create(AGenetic: TGenetic);
constructor CreateEmpty(AGenetic: TGenetic); virtual;
constructor Cross(AGenetic: TGenetic; Parent1, Parent2: TGen); virtual;
procedure FillStartData(AGenetic: TGenetic); virtual;
procedure Mutation(AGenetic: TGenetic); virtual;
procedure ComputeFitness(AGenetic: TGenetic); virtual;
property Fitness: Double read FFitness write SetFitness;
end;

The main methos for gen manipulation are CreateEmpty, Cross, FillStartData, Mutation,
ComputeFitness. The genetic algorithm uses this methods for computing. So if we can solve

some special problem with genetic algorithm, we have to override this methods. The genetic
algorithm constists from three basic stages
Evaluation of all gens
Cross some gens
Mutate some gens
With this three stages we can get new generation. Iteration is repeated multiple. After some
iteration the best gen (most height fitness) presents suboptimal solution. this iteration is
implemented as a method of class TGenetic.
type
TGenetic = class(TObject)
private
FGenerationCount: Longint;
FParentCount: Longint;
FGenerationSize: Longint;
FMutationPosibility: Longint;
FGenClass: TGenClass;
FGeneration: TGeneration;
FStatisticMutationCount: Longint;
procedure SetGenerationCount(const Value: Longint);
procedure SetGenerationSize(const Value: Longint);
procedure SetMutationPosibility(const Value: Longint);
procedure SetParentCount(const Value: Longint);
function GetBestGen: TGen;
function GetStatisticGenerationStage: Longint;
public
constructor Create;
destructor Destroy; override;

procedure OneGeneration;
procedure ResetGeneration;
property GenerationSize: Longint read FGenerationSize write
SetGenerationSize;
property GenerationCount: Longint read FGenerationCount write
SetGenerationCount;
property ParentCount: Longint read FParentCount write SetParentCount;
property MutationPosibility: Longint read FMutationPosibility write
SetMutationPosibility; {0-100}
property StatisticGenerationStage: Longint read
GetStatisticGenerationStage;
property StatisticMutationCount: Longint read FStatisticMutationCount;
property BestGen: TGen read GetBestGen;
property AcGeneration: TGeneration read FGeneration;
property GenClass: TGenClass read FGenClass write FGenClass;
end;

As I present above, the class TGenetic is encapsulation of genetic algorithm. Properties of class
allow setting parameters for genetic algorithm (GenerationSize, GenerationCount,
ParentCount, MutationPosibility). The iteration is implemented as method OneGeneration;
procedure TGenetic.OneGeneration;
var
I : Longint;
NewGeneration : TGeneration;
NewGen : TGen;
ParentIndex1 : Longint;
ParentIndex2 : Longint;
begin
if FGeneration = nil then
begin
FStatisticMutationCount := 0;
FGeneration := TGeneration.Create;
for I := 0 to GenerationSize - 1 do
begin
FGeneration.FGens.Add(FGenClass.Create(Self));
FGeneration[I].ComputeFitness(Self);
end;
end
else
begin
//move the best to new generation
NewGeneration := TGeneration.Create;
NewGeneration.FStage := FGeneration.FStage + 1;
I := 0;
while I < FParentCount do
begin
NewGeneration.FGens.Add(FGeneration.FGens[0]);
FGeneration.FGens.Delete(0);
Inc(I);
end;
//destroy old generation
FGeneration.Free;

FGeneration := NewGeneration;
//fill newgeneration with new childs
while I < FGenerationSize do
begin
ParentIndex1 := Random(FParentCount);
ParentIndex2 := ParentIndex1;
while ParentIndex1 = ParentIndex2 do
ParentIndex2 := Random(FParentCount);
//create new child
NewGen := FGenClass.Cross(Self, NewGeneration[ParentIndex1],
NewGeneration[ParentIndex2]);
NewGeneration.FGens.Add(NewGen);
if Random(100) < FMutationPosibility then
begin
//mutate gen
NewGen.Mutation(Self);
Inc(FStatisticMutationCount);
end;
NewGen.ComputeFitness(Self);
Inc(I);
end;
end;
//sort the population
FGeneration.SortByFitness;
end;

At first if there is not any generation, the first generation is created. If any generation exists new
generation is created. At second stage the new generation is filled by some better gens. At threed
stage are create new gens by crossing selected gens. The last stage select some gens and performs
some mutation. The last command in implementation is sorting of population.
Testing the framework
For testing we solve problem of function extrem computing. The class TGen was extend to
TExtremGen. Because we have to store some other information with genetic algorithm, we extend

class TGenetic too. The class TGenetic extension consist from information about the function
(criteria) and intervals.
type
TExtremGen = class(TGen)
private
FY: Double;
FX: Double;
public
constructor CreateEmpty(AGenetic: TGenetic); override;
constructor Cross(AGenetic: TGenetic; Parent1, Parent2: TGen); override;

procedure FillStartData(AGenetic: TGenetic); override;
procedure Mutation(AGenetic: TGenetic); override;
procedure ComputeFitness(AGenetic: TGenetic); override;
property X: Double read FX write FX;
property Y: Double read FY write FY;
end;

The gen stores the cooridnate of posible solution (X, Y). Gen behaviour is defined by overrided
methods.
constructor TExtremGen.Cross(AGenetic: TGenetic; Parent1, Parent2: TGen);
var
ExParent1 : TExtremGen absolute Parent1;
ExParent2 : TExtremGen absolute Parent2;
begin
FX := (ExParent1.X + ExParent2.X) / 2;
FY := (ExParent1.Y + ExParent2.Y) / 2;
end;
procedure TExtremGen.FillStartData(AGenetic: TGenetic);
begin
with AGenetic as TExtremGenetic do
begin
FX := MinX + (MaxX - MinX) * Random;
FY := MinY + (MaxY - MinY) * Random;
end;
end;
procedure TExtremGen.Mutation(AGenetic: TGenetic);
var
NX : Double;
NY : Double;
begin
with AGenetic as TExtremGenetic do
begin
NX := MinX - 1;
NY := 0;
while (NX < MinX) or (NX > MaxX) or (NY < MinY) or (NY > MaxY) do
begin
case Random(2) of
0: NX := FX + Random / 10 - 0.05;
1: NY := FY + Random / 10 - 0.05;
end;
end;
X := NX;
Y := NY;
end;
end;

Real results
The implemented algorithm are presented by real results from application

Pic. 1. First generation
In first generations gens are created randomly from defined intervals. Picture show 60 enities in
generation.

Pic. 2. Generation number 2
The genetic algorithm moves with points to global extrem in point (0, 0). The program show (on
left side in the middle) the actual results. Best gen evaluate current suboptimal solution (39.832e2) and the gen with worse fitness (43.035e+0). After some generations the problem is solved.

Pic. 3. Generation number 13
Last generation found the suboptimal solution in point F(0.001 ,0.000) = 18.386e-7.
Another function
The user (student) can choose some other function for evaluation. The current function is F(X,
Y) = 200 * Y * (Y + 2) / (X * X + X + 4);

Pic. 4. First generation

Pic. 5. The generation number 19.
The program found the extrem in point F(-0.499, -1.001)=-53.333e0.
Conclusion
We develop library that acts as general framework for genetic algorithms. For library testing was
implemented application for extrem function searching. This program is good for presenting the
genetic algorithm principles.

